PROC SQL: Make it a monster using these powerful functions and options  
Arun Raj Vidhyadharan, inVentiv Health, Somerset, NJ  
Sunil Mohan Jairath, inVentiv Health, Somerset, NJ

ABSTRACT

PROC SQL is indeed a powerful tool in SAS. However, we can make it even more powerful by using certain PROC SQL functions and options. This paper explores such functions and options that take PROC SQL to the next level.

INTRODUCTION

PROC SQL is often used in our day to day programming for its versatility and efficiency. An efficiently written PROC SQL can get us results much faster than a data step. It is also used for highly complex queries without the overhead of merging multiple datasets using data merge. However, there are certain rarely used functions and options that can be used with PROC SQL, which provides extra flexibility and coding power. Some of these functions are even undocumented. The purpose of this paper is to introduce these functions and options to PROC SQL users so that they can utilize the full potential of SQL coding.

The MONOTONIC function

The automatic variable _N_ in DATA step processing counts the number of times the DATA step begins to iterate. It’s very useful when you need the iteration number from the DATA step. Since PROC SQL uses a relational database concept that is different from the DATA step, we can’t get the iteration number from the PROC SQL procedure. The MONOTONIC function in PROC SQL can generate very similar result as the _N_ in DATA step. Look at the following example:

Consider the following dataset sashelp.class:

<table>
<thead>
<tr>
<th>Name</th>
<th>Sex</th>
<th>Age</th>
<th>Height</th>
<th>Weight</th>
</tr>
</thead>
<tbody>
<tr>
<td>Alfred</td>
<td>M</td>
<td>14</td>
<td>69</td>
<td>112.5</td>
</tr>
<tr>
<td>Alice</td>
<td>F</td>
<td>13</td>
<td>56.5</td>
<td>84</td>
</tr>
<tr>
<td>Barbara</td>
<td>F</td>
<td>13</td>
<td>65.3</td>
<td>98</td>
</tr>
<tr>
<td>Carol</td>
<td>F</td>
<td>14</td>
<td>62.8</td>
<td>102.5</td>
</tr>
<tr>
<td>Henry</td>
<td>M</td>
<td>14</td>
<td>63.5</td>
<td>102.5</td>
</tr>
<tr>
<td>James</td>
<td>M</td>
<td>12</td>
<td>57.3</td>
<td>83</td>
</tr>
<tr>
<td>Jane</td>
<td>F</td>
<td>12</td>
<td>59.8</td>
<td>84.5</td>
</tr>
<tr>
<td>Janet</td>
<td>F</td>
<td>15</td>
<td>62.5</td>
<td>112.5</td>
</tr>
<tr>
<td>Jeffrey</td>
<td>M</td>
<td>13</td>
<td>62.5</td>
<td>84</td>
</tr>
<tr>
<td>John</td>
<td>M</td>
<td>12</td>
<td>59</td>
<td>99.5</td>
</tr>
<tr>
<td>Joyce</td>
<td>F</td>
<td>11</td>
<td>51.3</td>
<td>50.5</td>
</tr>
</tbody>
</table>

The following PROC SQL will yield the result as shown below:
Example 1:

```
proc sql;
select monotonic() as rowno, *
from sashelp.class
where monotonic() le 5;
quit;
```

The above program will generate the output:

```
    rowno  Name      Sex       Age    Height    Weight
  -------------------------------
1  Alfred    M          14        69     112.5
2  Alice     F          13      56.5        84
3  Barbara   F          13      65.3        98
4  Carol     F          14      62.8     102.5
5  Henry     M          14      63.5     102.5
```

The MONOTONIC function is quite similar to the internal variable _N_ in DATA Step. We can use it to select the records according to their row number. For example, the following PROC SQL code chooses the observations from the 4th observation to the 8th observation in the sashelp.class dataset.

Example 2:

```
proc sql;
select * from sashelp.class where monotonic() between 4 and 8;
quit;
```

The COALESCE and COALESCEC functions

These functions return the first non-missing value from a list of arguments. COALESCE accepts one or more numeric arguments. The COALESCE function checks the value of each argument in the order in which they are listed and returns the first non-missing value. If only one value is listed, then the COALESCE function returns the value of that argument. If all the values of all arguments are missing, then the COALESCE function returns a missing value. The COALESCE function searches numeric arguments, whereas the COALESCEC function searches character arguments.

Consider the following dataset PATVISIT:

```
Figure 2.

<table>
<thead>
<tr>
<th>Patient</th>
<th>Visit1_Date</th>
<th>Visit2_Date</th>
<th>Visit3_Date</th>
<th>Visit4_Date</th>
<th>Visit5_Date</th>
</tr>
</thead>
<tbody>
<tr>
<td>100</td>
<td>8-Feb-14</td>
<td></td>
<td>15-Apr-14</td>
<td>21-May-14</td>
<td></td>
</tr>
<tr>
<td>101</td>
<td></td>
<td>12-Apr-14</td>
<td>9-Mar-14</td>
<td>19-Apr-14</td>
<td></td>
</tr>
<tr>
<td>102</td>
<td>4-Jan-14</td>
<td></td>
<td>16-Mar-14</td>
<td></td>
<td></td>
</tr>
<tr>
<td>103</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>104</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>105</td>
<td>2-Mar-14</td>
<td></td>
<td></td>
<td>12-May-14</td>
<td></td>
</tr>
</tbody>
</table>
```

If we want to find the first non-missing visit date for each patient, we could use the COALESCE function in PROC SQL as follows:

Example 3:

```
proc sql;
create table patvisit2 as select *, coalesce(visit1_date, visit2_date, visit3_date, visit4_date, visit5_date) as first_visit_date from patvisit;
quit;
```
The resulting dataset patvisit2 would look like this:

**Figure 3.**

<table>
<thead>
<tr>
<th>Patient</th>
<th>Visit1_Date</th>
<th>Visit2_Date</th>
<th>Visit3_Date</th>
<th>Visit4_Date</th>
<th>Visit5_Date</th>
<th>first_visit_date</th>
</tr>
</thead>
<tbody>
<tr>
<td>100</td>
<td>8-Feb-14</td>
<td>15-Apr-14</td>
<td>21-May-14</td>
<td>8-Feb-14</td>
<td></td>
<td></td>
</tr>
<tr>
<td>101</td>
<td></td>
<td>12-Apr-14</td>
<td>12-Apr-14</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>102</td>
<td></td>
<td>9-Mar-14</td>
<td>19-Apr-14</td>
<td></td>
<td>9-Mar-14</td>
<td></td>
</tr>
<tr>
<td>103</td>
<td>4-Jan-14</td>
<td>16-Mar-14</td>
<td></td>
<td></td>
<td>4-Jan-14</td>
<td></td>
</tr>
<tr>
<td>104</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>105</td>
<td></td>
<td>2-Mar-14</td>
<td>12-May-14</td>
<td>2-Mar-14</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

**The SPEDIS and SOUNDEX functions**

The SPEDIS function (stands for spelling distance) is used for fuzzy matching, which is comparing character values that may be spelled differently. The logic is a bit complicated, but using this function is quite easy. As an example, suppose you want to search a list of names to see if the name Friedman is in the list. You want to look for an exact match or names that are similar. Here is such a program:

**Using the SPEDIS function to perform a fuzzy match:**

**Example 4:**

```sql
data fuzzy;
  input Name $20.;
  Value = spedis(Name,'Friedman');
  datalines;
  Friedman
  Freedman
  Xriedman
  Freidman
  Friedmann
  Alfred
  FRIEDMAN
  ;
run;
```

Here is a listing of dataset Fuzzy:

**Figure 4.**

<table>
<thead>
<tr>
<th>Name</th>
<th>Value</th>
</tr>
</thead>
<tbody>
<tr>
<td>Friedman</td>
<td>0</td>
</tr>
<tr>
<td>Freedman</td>
<td>12</td>
</tr>
<tr>
<td>Xriedman</td>
<td>25</td>
</tr>
<tr>
<td>Freidman</td>
<td>6</td>
</tr>
<tr>
<td>Friedmann</td>
<td>3</td>
</tr>
<tr>
<td>Alfred</td>
<td>100</td>
</tr>
<tr>
<td>FRIEDMAN</td>
<td>87</td>
</tr>
</tbody>
</table>
The SPEDIS function returns a 0 if the two arguments match exactly. The function assigns penalty points for each type of spelling error. For example, getting the first letter wrong is assigned more points than misspelling other letters. Interchanging two letters is a relatively small error, as is adding an extra letter to a word.

Once the total number of penalty points has been computed, the resulting value is computed as a percentage of the length of the first argument. This makes sense because getting one letter wrong in a 3-letter word would be a more serious error than getting one letter wrong in a 10-letter word.

Notice that the two character values evaluated by the SPEDIS function are case-sensitive (look at the last observation in the listing). If case may be a problem, use the UPCASE or LOWCASE function before testing the value with SPEDIS.

To identify any name that is similar to Friedman, you could extract all names where the value returned by the SPEDIS function is less than some predetermined value. In the program here, values less than 15 or 20 would identify some reasonable misspellings of the name.

For human names, we can check similarities by the SOUNDEX function to avoid duplicates. The SASHELP.CLASS has 11 names. Phonically, John and Jane look similar according to the SOUNDEX function. Note that the SOUNDEX algorithm is English-biased and is less useful for languages other than English.

Example 5:
```
proc sql;
select a.name as name1, b.name as name2
from sashelp.class as a, sashelp.class as b
where soundex(name1) = soundex(name2) and (name1 gt name2);
quit;
```

SAS Output:
```
name1     name2
------------------
John      Jane
```

The IFC and IFN functions

The two functions play a role like the CASE-WHEN-END statements in typical SQL syntax, if the condition is about a binary selection. The IFC function deals with character variables, while the IFN function is for numbers. In the following example, IFN evaluates the expression TotalSales > 10000. If total sales exceed $10,000, then the sales commission is 5% of the total sales. If total sale is less than $10,000, then the sales commission is 2% of the total sales.

Example 6:
```
data Sales;
input TotalSales;
datalines;
25000
10000
500
10300
;
run;

proc sql;
select TotalSales,
ifn(TotalSales ge 10000, TotalSales*.05, TotalSales*.02) as commission from Sales;
quit;
```

The output:
COLON (:) MODIFIER COUNTERPART IN PROC SQL

In the DATA step, the colon operators (such as =:, >:, and <=:) can be used for truncated string comparisons. In PROC SQL, the following truncated string comparison operators are available: EQT, GTT, LTT, GET, LET and NET. These operators compare two strings after making the strings the same length by truncating the longer string to the same length as the shorter string. The truncation is performed internally. Neither operand is permanently changed. The following table lists the truncated comparison operators:

<table>
<thead>
<tr>
<th>Symbol</th>
<th>Definition</th>
<th>Example</th>
</tr>
</thead>
<tbody>
<tr>
<td>EQT</td>
<td>equal to truncated strings</td>
<td>where Name eqt 'Aust';</td>
</tr>
<tr>
<td>GTT</td>
<td>greater than truncated strings</td>
<td>where Name gtt 'Bah';</td>
</tr>
<tr>
<td>LTT</td>
<td>less than truncated strings</td>
<td>where Name ltt 'An';</td>
</tr>
<tr>
<td>GET</td>
<td>greater than or equal to truncated strings</td>
<td>where Country get 'United A';</td>
</tr>
<tr>
<td>LET</td>
<td>less than or equal to truncated strings</td>
<td>where Lastname let 'Smith';</td>
</tr>
<tr>
<td>NET</td>
<td>not equal to truncated strings</td>
<td>where Style net 'TWO';</td>
</tr>
</tbody>
</table>

JOIN METHOD CHOSEN BY PROC SQL

If you want to improve the join performance of programs that use PROC SQL to join tables, you need to know how the PROC SQL query optimizer chooses the join methods. There is an undocumented option METHOD on the PROC SQL statement that will display the hierarchy of processing methods that will be chosen by PROC SQL. You need to set the SAS System option MSGLEVEL = I to see the internal form of the query plan in the SAS LOG. The PROC SQL METHOD option can be used as an effective way to analyze a query process as well as for debugging purposes. An undocumented option TREE will show the hierarchy tree as planned in the SAS LOG.

The PROC SQL execution methods include:

- sqxcrta: Create table as Select
- sqxsdct: Select
- sqxjsl: Step Loop Join (Cartesian)
- sqxjim: Merge Join
- sqxjndx: Index Join
- sqxjhs: Hash Join
- sqxsor: Sort
- sqxsrc: Source Rows from table
- sqxtil: Filter Rows
- sqxsumg: Summary Statistics (with GROUP BY)
- sqxsumm: Summary Statistics (not grouped)
- sqxuniq: Distinct rows only
Example 7:

```sql
OPTIONS MSGLEVEL=I;
proc sql _METHOD_TREE;
select a.name, a.weight, b.predict
from sashelp.class a, sashelp.classfit b
where a.name=b.name;
quit;
```

The above program will display the query plan below in the SAS LOG file:

```
sqxsloc
  sqxjhsh
     sqxsloc( SASHELP.CLASS(alias = A) )
     sqxsloc( SASHELP.CLASSFIT(alias = B) )
```

Tree as planned.

```
  /SYM-V=(a.Name:1 flag=0001)
  /OBJ----|
  |      /SYM-V=(a.Weight:5 flag=0001)
  |      /SYM-V=(b.predict:6 flag=0001)
  /JOIN----|
  |      /SYM-V=(a.Name:1 flag=0001)
  |      /OBJ----|
  |      |      /SYM-V=(a.Weight:5 flag=0001)
  |      |      /SYM-V=(b.predict:6 flag=0001)
  |      /OBJ----|
  |      |      /SYM-V=(b.Name:1 flag=0001)
  |      /SRC----|
  |      \TABL[SASHELP].class opt=''
  |      /FROM----|
  |      |      /SYM-V=(b.predict:6 flag=0001)
  |      |      /SYM-V=(b.Name:1 flag=0001)
  |      /SYM-V=(b.Name:1)
  |      /CEQ----|
  |      \SYM-V=(b.Name:1)
```

**ALL, ANY, AND SOME**

ALL, ANY and SOME are special predicates that are oriented around subqueries. They are used in conjunction with relational operator. Actually, there are only two because ANY and SOME in SQL procedure are the same. ALL, ANY, and SOME are very similar to the IN predicate when it is used with subqueries; they take all the values produced by the subquery and treat them as a unit. However, unlike IN, they can be used only with subqueries. In SQL procedure, ALL and ANY differ from each other in how they react if the subquery procedures have no values to use in a
comparison. These differences can give your queries unexpected results if you do not account for them. One significant difference between ALL and ANY is the way they deal with the situation in which the subquery returns no values. In SQL procedure, whenever a legal subquery fails to produce output, the comparison operator modified with ALL is automatically TRUE, and the comparison operator modified with ANY is automatically FALSE.

Example 8:

```
Proc SQL;
select name, age from sashelp.class where age > ALL
(select age from sashelp.class where age >16);
select name, age from sashelp.class where age > ANY
(select age from sashelp.class where age >16);
select name, age from sashelp.class where age > SOME
(select age from sashelp.class where age >16);
quit;
```

The first query would produce the entire list of SASHELP.CLASS values as following results,

<table>
<thead>
<tr>
<th>Name</th>
<th>Age</th>
</tr>
</thead>
<tbody>
<tr>
<td>Alfred</td>
<td>14</td>
</tr>
<tr>
<td>Alice</td>
<td>13</td>
</tr>
<tr>
<td>Barbara</td>
<td>13</td>
</tr>
<tr>
<td>Carol</td>
<td>14</td>
</tr>
<tr>
<td>Henry</td>
<td>14</td>
</tr>
<tr>
<td>James</td>
<td>12</td>
</tr>
<tr>
<td>Jane</td>
<td>12</td>
</tr>
<tr>
<td>Janet</td>
<td>15</td>
</tr>
<tr>
<td>Jeffrey</td>
<td>13</td>
</tr>
<tr>
<td>John</td>
<td>12</td>
</tr>
<tr>
<td>Joyce</td>
<td>11</td>
</tr>
<tr>
<td>Judy</td>
<td>14</td>
</tr>
<tr>
<td>Louise</td>
<td>12</td>
</tr>
<tr>
<td>Mary</td>
<td>15</td>
</tr>
<tr>
<td>Philip</td>
<td>16</td>
</tr>
<tr>
<td>Robert</td>
<td>12</td>
</tr>
<tr>
<td>Ronald</td>
<td>15</td>
</tr>
<tr>
<td>Thomas</td>
<td>11</td>
</tr>
<tr>
<td>William</td>
<td>15</td>
</tr>
</tbody>
</table>

whereas the second and third queries would produce no output.

**DOUBLE option for Inserting a Blank Row into Output**

SQL can be made to automatically insert a blank row between each row of output. This is generally a handy feature when a single logical row of data spans two or more lines of output. By having SQL insert a blank row between each
logical record (observation), you create a visual separation between the rows — making it easier for the person reading the output to read and comprehend the output. The DOUBLE option is specified as part of the SQL procedure statement to insert a blank row and is illustrated in the following SQL code.

Example 9:
```
proc sql DOUBLE;
    select * from sashelp.cars;
quit;
```

CONCLUSION

Using the functions and options discussed above can drastically improve your programming approach and overall efficiency. These are added tools to your programming inventory that can be used in appropriate situations and hence avoid multiple data steps.

REFERENCES

Helpful Undocumented Features in SAS®
Wei Cheng, ISIS Pharmaceuticals, Inc., Carlsbad, CA

Top 10 Most Powerful Functions for PROC SQL
Chao Huang, Oklahoma State University
Yu Fu, Oklahoma State University

Handling Missing Values in the SQL Procedure
Danbo Yi, Abt Associates Inc., Cambridge, MA
Lei Zhang, Domain Solutions Corp., Cambridge, MA

A Hands-On Tour Inside the World of PROC SQL
Kirk Paul Lafler, Software Intelligence Corporation

ACKNOWLEDGMENTS

The authors would like to thank John Durski, Associate Director, inVentiv Health, for all his support and motivation in writing this paper.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Arun Raj Vidhyadharan
Enterprise: inVentiv Health
Address: 500 Atrium Drive
City, State ZIP: Somerset, NJ 08873
Work Phone: 732.652.3490
E-mail: arunraj.vidhyadharan@inventivhealth.com

Name: Sunil Mohan Jairath
Enterprise: inVentiv Health
Address: 500 Atrium Drive
City, State ZIP: Somerset, NJ 08873
Work Phone: 732.652.3482
E-mail: sunil.jairath@inventivhealth.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.